

The Difference between LA and LAY

The ability to load an address in assembler is a powerful tool. Working with an object from a distance (Load Address) rather than directly (Load) gives us the power of pointers and opens up the possibility of creating data structures. LA, an RX instruction, was the traditional instruction for working with addresses. LAY, an RXY-a instruction, offers some advantages over LA that you should consider.

RXY-a instructions come supplied with a 20-bit signed displacement compared to the 12-bit unsigned displacement of RX instructions. Consider what that means: With 20-bit signed displacements, the target address we are trying to access can be behind or in front of the base address. This opens up some new possibilities. Try it yourself with the small program below:

```
ASMSKEL  CSECT
          STM   14,12,12(13)
HERE      EQU   *
          BASR  12,0
          USING THERE,12
THERE     EQU   *
          ST    13,SAVEAREA+4
          LA    13,SAVEAREA
          LAY   5,HERE          REACH BACK TO ACCESS HERE
          DC    X'0000'
SAVEAREA  DS    18F
          END   ASMSKEL
```

When the program abends, look in register 5 and you will find the address of HERE. But HERE was defined before register 12's base address THERE. You couldn't get access to HERE using LA (try changing LAY to LA and running it again). The assembler complains that HERE is out of range. The reason is that LA is an RX instruction and the unsigned 12-bit displacements can only reach symbols in front of the base address THERE, but LAY does the trick. Pretty neat, eh?

Assembler programmers will sometimes use LA to increment a register value like this

```
LA      5,1(0,5)    INCREMENT R5 by 1
LA      5,L'X(0,5)  INCREMENT R5 BY THE LENGTH OF X
```

With LA, the displacements are limited to non-negative values from 0 to 4095. But LAY gives you more leeway:

```
LAY     5,20000(0,5)  INCREMENT R5 BY 20,000
LAY     5,-200(0,5)  DECREMENT R5 BY 200
```

With LAY, we can decrement as well as increment. Also with 20-bit signed displacements, LAY offers much larger displacements in the range -524,288 to +524,287.

I'm not suggesting that you abandon using LA – it still works fine. But be aware that LAY offers some advantages over LA, and the only disadvantage I can see is that it takes up six bytes in memory instead of the four bytes that LA uses. It's also the case that your machine has to have the long displacement facility installed.